Подключаюсь к стороннему веб-сервису. Имеется следующий фрагмент wsdl (см Снимок1). Перевожу на русский язык (в моей интерпретации).

Описан тип ParametersOfDocuments . Тип содержит единственное свойство ParametersOfDocument, имеющее множественное значение. Свойство ParametersOfDocument имеет тип BasicProperty, расширенный ещё двумя свойствами: Amount и Currency.

Нужно создать объект типа ParametersOfDocuments, заполнить его и передать в качестве параметра методу веб-сервиса. Код:

```1c
    Определение = Новый WSОпределения(...);
    Прокси = Новый WSПрокси(Определение, ...);

    ТипBasicProperty = Прокси.ФабрикаXDTO.Пакеты.Получить("...").Получить("BasicProperty");
    ТипParametersOfDocuments = Прокси.ФабрикаXDTO.Пакеты.Получить("...").Получить("ParametersOfDocuments");

    ТипCurrency = Прокси.ФабрикаXDTO.Пакеты.Получить("...").Получить("Currency");
    ТипCompany = Прокси.ФабрикаXDTO.Пакеты.Получить("...").Получить("Company");
    ТипBankAccountNum = Прокси.ФабрикаXDTO.Пакеты.Получить("...").Получить("BankAccountNum");

    Рубли = Прокси.ФабрикаXDTO.Создать(ТипCurrency);
    Рубли.Cod = Константы.ВалютаРегламентированногоУчета.Получить().Код;
    Рубли.Name = лев(Константы.ВалютаРегламентированногоУчета.Получить().Наименование, 3);

    Организация = Прокси.ФабрикаXDTO.Создать(ТипCompany);
    Организация.INN = "1234567890";
    Организация.KPP = "123456789";
    Организация.CNSI = "12345";

    БанковскийСчет = Прокси.ФабрикаXDTO.Создать(ТипBankAccountNum);
    БанковскийСчет.CNSI = "12345";
    БанковскийСчет.AccountNum = "12345678901234567890";
    БанковскийСчет.Country = "RU";

    BasicProperty = Прокси.ФабрикаXDTO.Создать(ТипBasicProperty);
    BasicProperty.Num = "1";
    BasicProperty.Date = ТекущаяДата();
    BasicProperty.Company = Организация;
    BasicProperty.BankAccountNum = БанковскийСчет;

    ПараметрыДокументов = Прокси.ФабрикаXDTO.Создать(ТипParametersOfDocuments);
    ParametersOfDocument = ПараметрыДокументов.Свойства().Получить(0);
    СписокParametersOfDocument = ПараметрыДокументов.ПолучитьСписок(ParametersOfDocument);
    СписокParametersOfDocument.Добавить(BasicProperty);
```

Последняя строка валится по ошибке:

Несоответствие типов XDTO:
Тип '{...}BasicProperty' не соответствует типу ''
Тип свойства не соответствует типу значения
{ВнешняяОбработка.ТестДоступаСКК.Форма.Форма.Форма(38)}: СписокParametersOfDocument.Добавить(BasicProperty);
{ВнешняяОбработка.ТестДоступаСКК.Форма.Форма.Форма(44)}: ПодключитьсяНаСервере();

по причине:
Несоответствие типов XDTO:
Тип '{...}BasicProperty' не соответствует типу ''
Тип свойства не соответствует типу значения

Как добавить в список XDTO значение расширенного типа?
---
Всем спасибо за участие! Выкладываю ответ, может кому пригодится.

```1c
    // Создаём объект XDTO типа ParametersOfDocuments
    ПараметрыДокументов = Прокси.ФабрикаXDTO.Создать(ТипParametersOfDocuments);
    // Получаем тип его первого (и единственного) свойства. Этот тип безымянный, поэтому по имени его получить не получается. Хотя мы знаем, что это ParametersOfDocument
    ТипParametersOfDocument = ПараметрыДокументов.Свойства().Получить(0).Тип;
    // Создаём объект XDTO типа ParametersOfDocument
    ParametersOfDocument = Прокси.ФабрикаXDTO.Создать(ТипParametersOfDocument);
    // Заполняем ео свойства, одноимённые со свойствами BasicProperty
    ЗаполнитьЗначенияСвойств(ParametersOfDocument, BasicProperty);
    // Заполняем расширенные свойства
    ParametersOfDocument.Amount = 1;
    ParametersOfDocument.Currency = Рубли;

    // Добавляем ParametersOfDocument в список
    // Получаем первое сойство объекта типа ParametersOfDocuments.
    СвойствоParametersOfDocument = ПараметрыДокументов.Свойства().Получить(0);
    // Получаем его список
    СписокParametersOfDocument = ПараметрыДокументов.ПолучитьСписок(СвойствоParametersOfDocument);
    // Добавляем объект типа ParametersOfDocument в список
    СписокParametersOfDocument.Добавить(ParametersOfDocument);
```
---
Возможно, здесь https://infostart.ru/1c/articles/716581/ получится найти что-то интересное по твоей теме
---
Всем спасибо за участие! Выкладываю ответ, может кому пригодится.

```1c
    // Создаём объект XDTO типа ParametersOfDocuments
    ПараметрыДокументов = Прокси.ФабрикаXDTO.Создать(ТипParametersOfDocuments);
    // Получаем тип его первого (и единственного) свойства. Этот тип безымянный, поэтому по имени его получить не получается. Хотя мы знаем, что это ParametersOfDocument
    ТипParametersOfDocument = ПараметрыДокументов.Свойства().Получить(0).Тип;
    // Создаём объект XDTO типа ParametersOfDocument
    ParametersOfDocument = Прокси.ФабрикаXDTO.Создать(ТипParametersOfDocument);
    // Заполняем ео свойства, одноимённые со свойствами BasicProperty
    ЗаполнитьЗначенияСвойств(ParametersOfDocument, BasicProperty);
    // Заполняем расширенные свойства
    ParametersOfDocument.Amount = 1;
    ParametersOfDocument.Currency = Рубли;

    // Добавляем ParametersOfDocument в список
    // Получаем первое сойство объекта типа ParametersOfDocuments.
    СвойствоParametersOfDocument = ПараметрыДокументов.Свойства().Получить(0);
    // Получаем его список
    СписокParametersOfDocument = ПараметрыДокументов.ПолучитьСписок(СвойствоParametersOfDocument);
    // Добавляем объект типа ParametersOfDocument в список
    СписокParametersOfDocument.Добавить(ParametersOfDocument);
```